Applied and Environmental Microbiology, Feb. 1982, p. 488–490 Vol. 43, No. 2
0099-2240/82/020488-03$02.00/0

# Estimation of the Most Probable Number with a Programable Pocket Calculator

ARTHUR L. KOCH
*Department of Biology, Indiana University, Bloomington, Indiana 47405*

Received 16 June 1981/Accepted 22 September 1981

The most-probable-number method has many potential applications, particularly if many tubes per dilution and many dilution levels are used. Increasing the number of cultures is possible with modern automatic and semiautomatic equipment. However, available tables are not sufficiently detailed to handle data from a large number of culture tubes used in an assay. This paper provides a computer program capable of handling the necessary arithmetic and written for a hand-held, advanced programable calculator.

The most-probable-number (MPN) method for the enumeration of bacteria and other objects is a powerful tool with a wide range of applications (3). It can be applied in some circumstances in which agar plate techniques will not work. With modern dispensing and mixing equipment, plastic ware, and optical scanners, enough cultures can be made so that the method is quite accurate (1, 3), but clearly, with a small number of tubes (three or five per level), the MPN method is not very accurate (5). In addition, the calculations involved are quite laborious. Consequently, the tables of Halvorson and Ziegler (2), made 50 years ago for 10 tubes per dilution, are usually consulted. These tables (and others) constrain experimental design in many ways, i.e., the number of levels, the volume of culture per tube, the number of tubes per level, and the factor of dilution between levels. The need for these tables has greatly decreased the utilization of the method and, unfortunately, has led to use of the method that fails to achieve maximum accuracy.

Additionally, these tables do not consider all of the possible outcomes for any determination. Thus, the tables of Halvorson and Ziegler for three levels of 10-fold dilutions with 10 tubes at each level only consider 210 outcomes of the possible 1,000. Of course, these tables consider the more likely possibilities, but if any of the 790 remaining possibilities occur, there is no way to interpret the data. Such odd combinations may represent an error or biological interference, but they may be real and cannot justifiably be rejected altogether.

With the advent of the programable pocket calculator, it is possible to write programs that do the necessary arithmetic for the MPN method. I have proposed a routine (3) that permits workers to dispense with tables if the simplest programable calculator is available. The computation must be carried out, however, in a trial-and-error fashion, but it allows a larger number of tubes per dilution to be used and thus permits accuracy that would not be possible with fewer tubes. However, the computation is somewhat difficult and requires considerable concentration. For this reason, I wrote a program to efficiently search for the MPN (for any experimental design) for a newly available, advanced programable calculator. Figure 1 is a program written for the hand-held Hewlett-Packard 41C (HP-41C) that searches for the MPN per unit volume of the culture of the first dilution submitted to the MPN assay. The same thing can be accomplished with a number of minicomputers or a large computer. The HP-41C is fast, and the search algorithm is efficient, but the computation takes 45 s to 4 min. I decided to write the program for this pocket calculator because it can be used almost anywhere.

To make the calculations convenient, the program can be applied to the conditions chosen by Halvorson and Ziegler without any special alterations. Any alterations from the default conditions may be done by changing the numbers stored in memories 08 through 11. Memory 08 currently contains 3, the number of levels; memory 09 contains 10, the dilution factor; memory 10 contains 10, the tubes per level; and memory 11 contains 10, the volume of culture used for the largest level. The program can be keyed into the calculator in 10 min and stored there or recorded on cards for ready reentry into the program memory.

With this flexibility, the method can be more readily adopted to any problem according to the considerations presented by Finney (1) and those previously discussed (3). In general, the total number of cultures should be as large as possible, and the number of levels and dilution factors should vary inversely with the expected

```
STATUS

USER
SIZE 021

                PRP "MPN"

 01♦LBL "MPN"
CLRG .1 STO 01 1
STO 02 3 STO 08 10
STO 09 STO 10 STO 11
STOP 16 ENTER↑ ENTER↑
RCL 08 + 1 - 1000 /
+ STO 12 STO 13 0 1

 28♦LBL 08
RCL 09 / ST+ Y ISG 13
GTO 08 RDN RCL 10 *
RCL 09 * RCL 11 *
STO 14 RCL 12 STO 13

 44♦LBL 09
"DATA" PROMPT
STO IND 13 ISG 13
GTO 09 GTO A

 51♦LBL 11
RCL 11 STO 03 RCL 14
STO 15 RCL 12 STO 13

 58♦LBL 10
RCL 03 ENTER↑ ENTER↑
RCL IND 13 * X<>Y
RCL 04 * CHS E↑X 1
- CHS / ST- 15
RCL 09 ST/ 03 ISG 13
GTO 10 RCL 15 RTN

 80♦LBL A
RCL 01 STO 04 XEQ 11
STO 05 RCL 02 STO 04
XEQ 11 STO 06 RCL 05
* X>0? GTO 05

 93♦LBL 00
RCL 02 RCL 02 RCL 01
- RCL 06 RCL 05 - /
RCL 06 * - STO 04
XEQ 11 STO 07 X=0?
GTO 04 ABS 1 E-4 X>Y?
GTO 04 RCL 07 RCL 06
* X>0? GTO 01 RCL 02
STO 01 RCL 06 STO 05

123♦LBL 02
RCL 04 STO 02 RCL 07
STO 06 GTO 00

129♦LBL 01
2 ST/ 05 GTO 02

133♦LBL 04
"MPN = " ARCL 04
PROMPT

137♦LBL 05
10 ST/ 01 ST* 02
GTO A END

EXAMPLE A

                XEQ "MPN"
                      RUN
DATA
          8.0000      RUN
DATA
          5.0000      RUN
DATA
          1.0000      RUN
MPN = 0.2676

EXAMPLE B
                XEQ "MPN"
          2.0000 STO 08
          5.0000 STO 09
         15.0000 STO 10
                      RUN
DATA
         12.0000      RUN
DATA
          3.0000      RUN
MPN = 0.1460
```

FIG. 1. Program and examples of MPN calculations. The program basically calculates the quantity $\Sigma[(\alpha_i\rho_i)/(1 - e^{-\alpha_i x})]$, where $x$ is an estimate of MPN. The number of positive cultures in the $i^{th}$ dilution is $\rho_i$, and the original volume tested is $\alpha_1$. This summation equals $\Sigma\alpha_i\eta_i$, where $\eta_i$ is the number of total tests performed at the $i^{th}$ dilution. When $x$ has been properly chosen, the difference in the two summations will be zero (2). Under the default conditions achieved by keying "ln" and "R/S," the number of levels (maximum value of $i$) is set at 3 in memory 08; the dilution factor $\alpha_1/\alpha_{1+1}$ is set at 10 in memory 09; the number of tubes per level $\eta_i$ is set at 10 in memory 10; and the volume of culture for each test of the least diluted sample, $\alpha_1$ is set at 10 ml in memory 11. The program assumes that the dilution factor and the number of tubes per level are constant. At step 01, the program labeled "MPN" stores these values and certain bookkeeping constants, such as the assumption that $x$ lies between 0.1 and 1. At step 44, the program labeled 09 asks the operator for the $i$ successive numbers of positive cultures at the different levels. Starting at step 28, the program stores the second summation. Then at step 51, program 11 computes the first summation and subtracts the second. The difference of the two sums is a measure of how far away the value of $x$ is from the correct one. Program 11 is the heart of the computation and serves as a subroutine, starting at step 80, which takes the initial range of $x$ and tests to determine whether the correct value is between 0.1 and 1. If not, the program branches to step 137, where the minimum is decreased 10-fold and the maximum is increased 10-fold, and the calculator tries again. If $x$ is within the range, a better value is estimated from $x_2 - (x_2 - x_1) D_2/(D_2 - D_1)$, where $x_1$ and $x_2$ are initially 0.1 and 1, $D_1$ is the value of difference of summations for $x = 0.1$, and $D_2$ is the value for $x = 1$. If the new value corresponds to a value of $D$ which is less than $10^{-4}$, the computation is over and the calculator shows (or prints) the MPN. If not, the new value of $x$ becomes $x_2$, but depending whether the answer lies between the new value of $x$ and $x_1$ or the previous $x_2$ value, either that value becomes $x_1$, or $D_1$ is replaced by $D_1/2$, and $x_1$ remains at the same value. In either case, the program returns to step 93. With this program, the low-memory version of the HP-41C suffices. With size set 021, six memory registers are devoted to the storage of data, and six levels of dilution can be analyzed. This is probably larger than will be needed even when small dilution factors, e.g. twofold, are used, but 19 dilution levels can be handled without additional memory.

accuracy of the predicted number. Thus, when users are confident of the estimated value, they can use fewer levels and smaller dilution factors than when a poor or no estimate is available. For example, for the microtiter plate technique of Rowe et al. (4), the plate should be run with twofold dilutions with eight wells per dilution when the titer is uncertain. It should be twisted

90° and run with 12 wells per twofold dilution when an accurate estimate is available.

Figure 1 (example A) presents an example that has been previously considered (3). Also presented is an example in which all but one of the constraints are different from the default values. When in operation, the calculator is placed in the user mode, program MPN is assigned to "ln," and the upper right-hand button (marked "ln") is pushed. Once the calculator stops, any alterations in the default values are stored in the appropriate memories. The number of positive cultures for the first dilution is then keyed in. After this has been done, the lower "R/S" button is pushed. The calculator then solves and displays the MPN. Although the calculator normally displays the answer to four decimals, the error of the method warrants fewer decimals. Graphs showing the precision have been described previously (3). The output in Fig. 1 was produced on an associated printer in the "norm" mode.

This work was supported by National Science Foundation grant PCM 79-1141.

## LITERATURE CITED


1. **Finney, D. J.** 1964. Statistical methods in biological assay, 2nd ed., p. 570–586. Hafner Press, New York.
2. **Halvorson, H. O., and N. R. Ziegler.** 1933. Application of statistics to problems in bacteriology. I. A means of determining bacterial population by the dilution method. J. Bacteriol. **25**:101–121.
3. **Koch, A. L.** 1981. Growth measurement, p. 179–207. *In* P. Gerhardt, R. G. E. Murray, R. N. Costilow, E. W. Nester, W. A. Wood, N. R. Krieg, and G. B. Phillips (ed.), Manual of methods for general bacteriology. American Society for Microbiology, Washington, D.C.
4. **Rowe, R., R. Todd, and J. Waide.** 1977. Microtechnique for most-probable-number analysis. Appl. Environ. Microbiol. **33**:675–680.
5. **Woodward, R. L.** 1957. How probable is the most probable number? J. Am. Water Works Assoc. **49**:1060–1068.